

Introducción a la Programación Funcional

Nota de clase 03: Recursión y Funciones de Orden Superior*

Karla Ramírez Pulido

Manuel Soto Romero

22 de enero de 2020

Una de las principales características de los lenguajes funcionales, como RACKET es el uso de la recursión como método de repetición en funciones. Se dice la definición de una función es recursiva cuando se hace uso de ésta en la definición misma, a esto se le conoce como autoreferencia. Una función recursiva *válida* consta de dos partes:

- Un conjunto de casos base, los cuales son casos simples donde la definición se da directamente, es decir, sin usar autoreferencia.
- Un conjunto de reglas recursivas donde se define un nuevo caso de la definición en términos de anteriores ya definidos.

A continuación se presentan algunas funciones sobre listas, para manipular las mismas se hace uso de recursión.

3.1. Recursión sobre listas

Ejemplo 3.1. Definir una función `longitud` tal que `(longitud l)` representa la longitud de una lista. Por ejemplo:

```
(longitud '()) = 0
(longitud '#\a)) = 1
(longitud '(5.0 1.3 2.7)) = 3
```

Solución

```
1 ;; Función que obtiene la longitud de una lista.
2 ;; longitud: (listof a) $ \color{Verde}\rightarrow$ number
3 (define (longitud l)
4   (if (empty? l)
5       0
6       (+ 1 (longitud (cdr l)))))
```

Código 1: Longitud de una lista

□

Ejemplo 3.2. Definir una función `quita` tal que `(quita n l)` representa a la lista `l` sin sus primeros `n` elementos. Por ejemplo:

```
(quita 0 '(1 2 3)) = '(1 2 3)
(quita 2 '#\H #\o #\l #\a)) = '#\l #\a)
```

Solución

* Adaptación de las Notas de Laboratorio del Curso de Lenguajes de Programación.

```

1 ;; Función que quita los primeros n elementos de una lista.
2 ;; quita: number (listof a) → (listof a)
3 (define (quita n l)
4   (if (zero? n)
5       l
6       (quita (sub1 n) (cdr l))))

```

Código 2: Lista sin los primeros n elementos

□

Ejemplo 3.3. Definir una función `toma` tal que `(toma n l)` representa los primeros `n` elementos de una lista `l`. Por ejemplo:

```

(toma 0 '(1 2 3))      = '()
(toma 2 '(\H \o \l \a)) = '(\H \o)

```

Solución:

```

1 ;; Función que toma los primeros n elementos de una lista.
2 ;; toma: number (listof a) → (listof a)
3 (define (toma n l)
4   (if (zero? n)
5       '()
6       (cons (car l) (toma (sub1 n) (cdr l)))))

```

Código 3: Primeros n elementos de una lista

□

Ejemplo 3.4. Definir una función `contiene` tal que `(contiene e l)` indica si el elemento `e` es elemento de la lista `l`. Por ejemplo:

```

(contiene? 0 '(1 2 3))      = #f
(contiene? 2 '())           = #f
(contiene? #\o '(\H \o \l \a)) = #t

```

Solución:

```

1 ;; Función que indica si un elemento pertenece a una lista.
2 ;; contiene: a (listof a) → boolean
3 (define (contiene? e l)
4   (or (equal? (car l) e) (contiene? e (cdr l))))

```

Código 4: Indica si un elemento pertenece a una lista

□

3.2. Funciones de orden superior

Las funciones de RACKET actúan como cualquier otro valor en el lenguaje, por ejemplo, pueden pasarse como parámetro a otras funciones. Existen algunas funciones de este tipo para trabajar con listas, entre las que se encuentran: `map` y `filter` y las funciones de plegado¹ `foldr` y `foldl`. A continuación se muestra la implementación de estas funciones y algunos ejemplos de uso.

¹Pertenecientes a la llamada programación origami.

3.2.1. Función map

Esta función aplica a cada elemento de una lista la función que recibe como parámetro. La definición de esta función se aprecia en el Código 5.

```
1 ;; Función que aplica una función a cada elemento de una lista.
2 ;; map: procedure (listof any) → (listof any)
3 (define (map f l)
4   (match l
5     ['() '()]
6     [(cons x xs) (cons (f x) (map f xs))]))
```

Código 5: Implementación de map

Ejemplo 3.5. Definir una función `meses` tal que `(meses l)` transforma una lista de números enteros en una lista de cadenas que representan el mes correspondiente. Por ejemplo:

```
(meses '()) = '()
(meses '(10 12)) = '("Octubre" "Diciembre")
```

Solución

```
1 ;; Función que obtiene el nombre del mes representado por el número
2 ;; recibido como parámetro.
3 ;; nombre-mes: number → string
4 (define (nombre-mes n)
5   (cond
6     [(equal? n 1) "Enero"]
7     [(equal? n 2) "Febrero"]
8     [(equal? n 3) "Marzo"]
9     [(equal? n 4) "Abril"]
10    [(equal? n 5) "Mayo"]
11    [(equal? n 6) "Junio"]
12    [(equal? n 7) "Julio"]
13    [(equal? n 8) "Agosto"]
14    [(equal? n 9) "Septiembre"]
15    [(equal? n 10) "Octubre"]
16    [(equal? n 11) "Noviembre"]
17    [(equal? n 12) "Diciembre"]
18    [else (error 'nombre-mes "Mes inválido.")]))
19
20 ;; Función que transforma una lista de números enteros en una lista
21 ;; de cadenas que representan el mes correspondiente.
22 ;; meses: (listof number) → (listof string)
23 (define (meses l)
24   (map nombre-mes l))
```

Código 6: Función meses

La función `meses` consiste de aplicar `nombre-mes` a cada elemento de la lista recibida como parámetro.

□

3.2.2. Función filter

Esta función recibe un predicado y deja en la lista aquellos elementos que cumplan con el mismo. La definición de esta función se aprecia en el Código 7.

```

1 ;; Función que aplica filtra los elementos de una lista dada una
2 ;; condición.
3 ;; filter: (any -> boolean) (listof any)  $\rightarrow$  (listof any)
4 (define (filter f l)
5   (match l
6     ['() '()]
7     [(cons x xs)
8       (cond
9         [(f x) (cons x (filter f xs))]
10        [else (filter f xs)])]))

```

Código 7: Implementación de filter

Ejemplo 3.6. Definir una función `ternas-pitagoricas` tal que `(ternas-pitagoricas l)` filtra listas de longitud 3 que cumplen con la propiedad de ser una terna pitagórica. Por ejemplo:

```

(ternas-pitagoricas '()) = '()
(ternas-pitagoricas '((1 2 3) (3 4 5))) = '((3 4 5))

```

Solución:

```

1 ;; Función que indica si una lista de tamaño 3 es una terna pitagórica.
2 ;; terna-pitagorica?: list  $\rightarrow$  boolean
3 (define (terna-pitagorica? l)
4   (match l
5     ['() #f]
6     [(list u v w) (equal? (+ (expt u 2) (expt v 2)) (expt w 2))])
7
8 ;; Función que filtra las tuplas de tamaño 3 que cumplen con la
9 ;; propiedad de ser una terna pitagórica.
10 ;; ternasPitagoricas: (listof list)  $\rightarrow$  boolean
11 (define (ternas-pitagoricas xs)
12   (filter terna-pitagorica? xs))

```

Código 8: Filtra listas de longitud 3 que cumplen con la propiedad de ser una terna pitagórica

Se hace uso del predicado `terna-pitagorica?` definida en las líneas 3 a 6. De esta forma, la función `ternas-pitagoricas` consiste de filtrar usando la función `terna-pitagorica?` cada elemento de la lista recibida como parámetro.

□

3.2.3. Funciones foldr y foldl

Estas funciones engloban un patrón es común, que consiste en aplicar una función mediante un recorrido de derecha izquierda (`foldr`) o de izquierda a derecha (`foldl`).

Estas funciones requieren de tres parámetros:

1. Una función a aplicar.
2. Un valor a regresar cuando se tenga un caso base (lista vacía).
3. La estructura sobre la cual se realizará el recorrido (lista).

A continuación se presenta la implementación de estas dos funciones (sobre listas):

```

1 ;; Función que aplica una función a los elementos de una lista, de
2 ;; forma encadenada a la derecha.
3 ;; foldr: procedure any (listof any)  $\rightarrow$  any
4 (define (foldr f v l)
5   (match l
6     ['() v]
7     [(cons x xs) (f x (foldr f v xs))]))
8
9 ;; Función que aplica una función a los elementos de una lista, de
10 ;; forma encadenada a la izquierda.
11 ;; foldl: procedure any (listof any)  $\rightarrow$  any
12 (define (foldl f v l)
13   (match l
14     ['() v]
15     [(cons x xs) (foldl f (f x v) xs)]))

```

Código 9: Implementaciones de foldr y foldl

Ejemplo 3.7. Definir dos funciones suma-listar y suma-listal tal que (suma-listar l) y (suma-listal l) suman los elementos de una lista a la derecha e izquierda respectivamente. Por ejemplo:

```

(suma-listar '())      = 0
(suma-listal '())      = 0
(suma-listar '(1))     = 1
(suma-listal '(1))     = 1
(suma-listar '(1 2 3)) = 6
(suma-listal '(1 2 3)) = 6

```

Solución:

```

1 ;; Función que suma los elementos de una lista.
2 ;; suma-listar: (listof number)  $\rightarrow$  number
3 (define (suma-listar xs)
4   (foldr + 0 xs))
5
6 ;; Función que suma los elementos de una lista.
7 ;; suma-listal: (listof number)  $\rightarrow$  number
8 (define (suma-listal xs)
9   (foldl + 0 xs))

```

Código 10: Suman los elementos de una lista

La Figura 1, muestra, a la izquierda, el árbol de evaluación de la llamada a `suma-listar` con la lista `'(1 2 3)` y del lado derecho a `suma-listal` con la misma entrada.




Figura 1: Árboles de evaluación de suma-listar y suma-listal

□

3.2.4. Funciones anónimas (lambdas)

Otro tipo de función importante, en Racket, son las *lambdas*. Este tipo de funciones son útiles cuando se desea usar una función con alcance local, esto es, que no esté disponible en todo el archivo de definiciones, sino únicamente dentro de la expresión donde son usadas. Debe su nombre a las funciones del Cálculo Lambda de Alonzo Church.

Son utilizadas principalmente en combinación con otras funciones de orden superior como son `map` y `filter`, para realizar aplicaciones o filtros de funciones que no se volverán a usar. La sintaxis de una lambda es alguna de las siguientes:

```
(lambda (<parámetro1> ... <parámetroN>)
  <cuerpo>)
```

```
(λ (<parámetro1> ... <parámetroN>)
  <cuerpo>)
```

No se provee ningún nombre para la lambda, pues es una función anónima, se separan los parámetros por espacios y se indica un cuerpo. A continuación se presentan algunos ejemplos de uso de lambdas.

```
> ((lambda (x y) (+ x y)) 17 29)
46
> (map (λ (x) (+ x 13)) '(1 2 3))
'(14 15 16)
> (filter (lambda (x) (zero? (modulo x 13))) '(1 2 13))
'(13)
```

Referencias

- [1] Favio Miranda, Elisa Viso, *Matemáticas Discretas*, Las Prensas de Ciencias, Segunda Edición, 2015.
- [2] Matthias Felleisen, Conrad Barski, et.al., *Realm of Racket*, No Starch Press, Primera Edición, 2013.
- [3] Matthias Felleisen, Robert B. Findler, et.al., *How to Design Programs*, MIT Press, Segunda Edición, 2019.
- [4] Matthew Flatt, Robert B. Findler, et. al., *The Racket Guide*, Versión 7.5
- [5] Éric Tanter, *PrePLAI: Scheme y Programación Funcional*, 2014.